



# ML 演習 第 2 回

---

おおいわ  
April 16, 2002

# 今回の内容

- パターンマッチ
- Parametric Polymorphism
  - リスト
  - 多相関数
  - ユーザ定義データ型
    - レコード型
    - バリエント型
  - 多相データ型と多相関数

# パターンマッチ (1)

- 入力データとパターンを照合する

```
# let rec fib v = match v with  
  | 0 | 1 -> 1  
  | x -> fib(x-1) + fib(x-2);;  
val fib : int -> int = <fun>
```

パターン

# パターンマッチ (2)

## ■ 関数定義におけるパターンマッチ

```
# let f (x,y) = x + y;;  
val f : int * int -> int = <fun>  
# f (2,3);;  
- : int = 5
```

# パターンマッチ (3)

- 様々なパターン
  - 整数
  - ペア
  - リスト・ユーザ定義型など (後述)
  - これらの複雑な組み合わせ

# パターンマッチ (4)

## ■ マッチできない値のある場合の警告

```
# let flip x = match x with  
  0 -> 1 | 1 -> 0;;
```

```
Warning: this pattern-matching is not exhaustive.  
Here is an example of a value that is not matched:  
2
```

```
val flip: int -> int
```

```
# flip 2;;
```

```
Uncaught exception: Match_failure ("", 13, 41).
```

# リスト (1)

## ■ リストの基本要素

### ■ 空リスト

```
# [] ;;
```

```
- : 'a list = []
```

### ■ cons コンストラクタ

```
# 1 :: [] ;;
```

```
- : int list = [1]
```

```
# 1 :: (2 :: (3 :: [])) ;;
```

```
- : int list = [1; 2; 3]
```

# リスト (2)

## ■ リストの操作 (1): cons と append

```
# let l1 = [1; 2; 3; 4; 5];;
```

```
val l1 : int list = [1; 2; 3; 4; 5]
```

```
# -1 :: 0 :: l1;;
```

```
- : int list = [-1; 0; 1; 2; 3; 4; 5]
```

```
# l1 @ [6; 7];;
```

```
- : int list = [1; 2; 3; 4; 5; 6; 7]
```



# リスト (3)

## ■ リストの分解: パターンマッチを使う

```
# let rec sum l = match l with  
    [] -> 0  
    | hd :: tl -> hd + sum tl
```

```
val sum : int list -> int = <fun>
```

```
# sum [1; 2; 3; 4; 5];;
```

```
- : int = 15
```



1 :: [2; 3; 4; 5]

# 型多相とは (1)

- 例1: リストの先頭要素を得る操作 `hd`
  - `int_hd: int list -> int`
  - `bool_hd: bool list -> bool`
  - `string_hd: string list -> string`
  - `intpair_hd: (int * int) list -> int * int`
  - etc...
  - 操作は共通: `let hd (h::t) = h`  
共通な定義は与えられない?

# 型多相とは (2)

- 解決: 「型」についてパラメータ化する
  - $\text{hd}[\alpha] : \alpha \text{ list} \rightarrow \alpha$ 
    - $\text{hd}[\text{int}] : \text{int list} \rightarrow \text{int}$
    - $\text{hd}[\text{string}] : \text{string list} \rightarrow \text{string}$
    - $\text{hd}[\text{int} * \text{bool}] : (\text{int} * \text{bool}) \text{ list} \rightarrow \text{int} * \text{bool}$
  - ML では、 $[\alpha]$  の部分は明示しなくてよい (型推論で自動的に解決)

# 多相関数 (1)

## ■ 例1: 恒等関数

```
# let id x = x;;  
id : 'a -> 'a = <fun>  
# id 1;;  
- : int = 1;;  
# id [true; false];;  
- : bool list = [true; false]  
# id sum;;  
- : int list -> int = <fun>
```

型変数  $\alpha$  などは  
'a, 'b, ... と表記する

# 多相関数 (2)

- 例2: `fst, snd` : ペアの要素取り出し

```
# let fst (x, _) = x;;
```

```
val fst : 'a * 'b -> 'a = <fun>
```

```
# let snd (_, y) = y;;
```

```
val snd : 'a * 'b -> 'b = <fun>
```

*cf.* `_`: 何とでもマッチする匿名パターン

- 適当な名前を考える代わりに `_` を使い  
計算に使わないことを明示できる

# 多相関数 (3)

## ■ 例3: mk\_list

```
# let rec mk_list n v =  
    if n = 0 then []  
    else v :: mk_list (n-1) v;;  
val mk_list: int -> 'a -> 'a list  
# mk_list 3 "1";;  
- : string list = ["1"; "1"; "1"]  
# mk_list 2 true;;  
- : bool list = [true; true]
```

# 多相関数 (4)

## ■ 例3: rev : リストの反転

```
# let rev l =  
    let rec iter s d = match s with  
        [] -> d  
        | (h::t) -> iter t (h::d)  
    in iter l []  
  
val rev: 'a list -> 'a list  
# rev [1; 2; 3];;  
- : int list = [3; 2; 1]
```

# 多相関数 (4')

## ■ 例3: rev (解説)

```
rev [1; 2; 3]
⇒ iter [1; 2; 3] []
⇒ iter [2; 3] [1]
⇒ iter [3] [2; 1]
⇒ iter [] [3; 2; 1]
⇒ [3; 2; 1]
```

```
let rec rev l =
  let rec iter s d =
    match s with
    [] -> d
    | (h::t) ->
      iter t (h::d)
  in iter l []
```



# 多相關数 (5)

## ■ 例4: : map (高階関数)

```
# let rec map f l = match l with
  [] -> []
  | hd :: tl -> f hd :: map f tl;;
val map : ('a -> 'b) ->
  'a list -> 'b list = <fun>
# map fib [1; 2; 3; 4; 5];;
- : int list = [1; 1; 2; 3; 5]
```

# 多相関数 (6)

## ■ 型の明示的な制限

```
# let f1 x = (x, x);;
val f1 : 'a -> 'a * 'a = <fun>
# let f2 (x:int) = (x, x);;
val f2 : int -> int * int = <fun>
# let f3 x = ((x, x):int * int);;
val f3 : int -> int * int = <fun>
# f3 "string"
This expression has type string ...
```

# 独自データ型の定義

- レコード (record)
  - 複数の値の組の型
    - C 言語の struct に相当
- バリエーション (variant)
  - 複数の値の種類のうち1つを値とする型
    - C 言語の enum, union, cast などの組み合わせに相当
    - 操作の安全性をきちんと保証してくれる

# レコード型 (1)

## ■ 例: 複素数の直交座標表示

```
# type complex =  
    {re : float; im : float};;  
type complex =  
    { re : float; im : float; }  
# let c1 = {re = 5.0; im = 3.0};;  
val c1 : complex  
    = {re=5.000000; im=3.000000}  
# c1.re;;  
- : float = 5.000000
```

# レコード型 (2)

## ■ パターンマッチング

```
# let add_comp
    {re=r1; im=i1} {re=r2; im=i2}
  = {re = r1 +. r2; im = i1 +. i2};;

val add_comp = complex -> complex
                -> complex = <fun>

# add_comp c1 c1;;

- : complex
  = {re=10.000000; im=6.000000}
```

# バリエアント型 (1)

## ■ 例1: 整数をノードにもつ木

```
# type itree = Leaf
  | Node of int * itree * itree;;
type itree = Leaf
  | Node of int * itree * itree
# Leaf;;
- : itree = Leaf
# Node(5,Leaf,Leaf);;
- : itree = Node (5, Leaf, Leaf)
```

# バリエアント型 (2)

## ■ 木のノードの値の合計を求める関数

```
# let rec sum_itree t = match t with
  Leaf -> 0   | Node(a,t1,t2) ->
    a + sum_itree t1 + sum_itree t2;;
val sum_itree : itree -> int = <fun>
# sum_itree Leaf;;
- : int = 0
# sum_itree
  (Node(4,Node(5,Leaf,Leaf),Leaf));;
- : int = 9
```

# 多相データ型とは (1)

## ■ 例2: 一般の「木」とは?

- $\text{itree} = \text{Leaf} \mid \text{Node of } \text{int} * \text{itree} * \text{itree}$
- $\text{btree} = \text{Leaf} \mid \text{Node of } \text{bool} * \text{btree} * \text{btree}$
- $\text{ibtree} = \text{Leaf} \mid$   
 $\text{Node of } (\text{int} * \text{bool}) * \text{ibtree} * \text{ibtree}$
- 多相関数と同じような考え方が出来ないか?



# 多相データ型とは (2)

## ■ 再び「型に関するパラメータ化」

### ■ 一般型

■  $\alpha$  tree = Leaf | Node of  $\alpha$  \*  $\alpha$  tree \*  $\alpha$  tree

■ int tree = Leaf | Node of int \* int tree \* int tree

■ bool tree = Leaf |  
Node of bool \* bool tree \* bool tree

■ ...

# 多相データ型 (1)

## ■ 例2: 要素をノードにもつ木

```
# type 'a tree = Leaf | Node of
    'a * 'a tree * 'a tree;;
type 'a tree = Leaf | Node of
    'a * 'a tree * 'a tree
# Node(5, Leaf, Leaf);;
- : int tree = Node (5, Leaf, Leaf)
# Node("ocaml", Leaf, Leaf);;
- : string tree =
    Node ("ocaml", Leaf, Leaf)
```

# 多相データ型 (2)

- システム組み込みの多相バリエーション型
  - $\alpha$  option = None | Some of  $\alpha$ 
    - 「ないかもしれない」データを表す型
    - C では NULL ポインタの利用が一般的
  - $\alpha$  list = [] | :: of  $\alpha$  \*  $\alpha$  list
    - 構文的にはちょっと特殊

# 多相データ型と多相関数

## ■ (例) tree の深さを計算する関数

```
# let rec depth t = match t with
  Leaf -> 0 | Node(_,t1,t2) ->
  let d1 = depth t1 in
  let d2 = depth t2 in
  1 + (if d1>d2 then d1 else d2);;

val depth : 'a tree -> int = <fun>
# depth(Node(5,Node(4,Leaf,Leaf),Leaf));;
- : int = 2;
```

# その他の構文 (1)

## ■ 関数を作る構文 (1)

- fun(, let) → 多引数関数と1つのパターン

```
# (fun x y -> x + y) 2 3;;
```

```
- : int = 5
```

```
# let f x y = x + y;;
```

```
val f : int -> int -> int = <fun>
```

```
# f 2 3;;
```

```
- : int = 5
```

# その他の構文 (2)

## ■ 関数を作る構文 (2)

- function: 1引数だが複数パターン

```
# let isnull = function
    [] -> true | _ -> false;;
val isnull : 'a list -> bool = <fun>
# let rec map f = function
    [] -> []
    | hd::tl -> f hd :: map f tl;;
val map : ('a -> 'b) ->
    'a list -> 'b list = <fun>
```

# その他の構文 (3)

## ■ コメント

### ■ (\* と \*) の間

```
# 1 + (* this is comment *) 2;;
```

```
- : int 3
```

### ■ 入れ子にできる

```
# 1 + (* 2 + (* 3 + *) 4 + *) 5;;
```

```
- : int = 6
```

# その他の構文 (4)

- 中置演算子について
    - 実体は通常の2変数関数
    - `()` で括ると通常値として使える
- ```
# ( * );;  
- : int -> int -> int = <fun>  
  
# ( * ) 5 3;;  
- : int = 15
```



# 課題1

- 引数の2リストを連結したリストを返す関数 `append:  $\alpha$  list  $\rightarrow$   $\alpha$  list  $\rightarrow$   $\alpha$  list` を `@` を使わずに定義せよ。

```
# append [1; 2] [3; 4; 5];;
```

```
- : int list = [1; 2; 3; 4; 5]
```

# 課題2

- 判定関数とリストを受け取り、元のリストの要素のうち条件を満たす要素だけからなるリストを生成する関数  
 $\text{filter}: (\alpha \rightarrow \text{bool}) \rightarrow \alpha \text{ list} \rightarrow \alpha \text{ list}$   
を定義せよ。


```
# filter odd [1; 2; 3; 4; 5];;  
- : int list = [1; 3; 5]
```

# 課題3

- 木 ( $\alpha$  tree) を受け取り深さ優先探索で、全要素を並べたリストを生成する関数  $\text{dfs}: \alpha \text{ tree} \rightarrow \alpha \text{ list}$  を定義せよ。

# 課題3 (例)

```
# dfs(Node(7,  
Node(5,Node(4,Leaf,Leaf),Leaf),  
Node(9,Leaf,Node(15,Leaf,Leaf))));;  
- : int list = [7; 5; 4; 9; 15]
```



# 課題4 (optional)

- 2項演算子  $\oplus$  と零元  $z$  とリスト  $[a_1; a_2; \dots; a_n]$  を受け取り、右結合で結合させた結果  $a_1 \oplus (a_2 \oplus (\dots \oplus (a_n \oplus z) \dots))$  を返す関数  $\text{foldr}: (\alpha \rightarrow \beta \rightarrow \beta) \rightarrow \beta \rightarrow \alpha \text{ list} \rightarrow \beta$  を定義せよ。

```
# foldr (+) 0 [1; 2; 3; 4; 5];;
```

```
- : int = 15
```

# 課題4 (例)

```
# let flatten x = foldr (@) [] x;;
flatten : 'a list list -> 'a list
# flatten [[1;2]; [3;4]; [5;6;7]];;
- : int list = [1; 2; 3; 4; 5; 6; 7]
# let filter f x = foldr
  (fun x y -> if f x then x::y else y)
  [] y;;
filter : 'a list -> 'a list
```

# 課題5 (optional)


- 課題3 (例) の木は、左の枝に含まれる要素はノードの値より小さく、右の枝に含まれる要素はノードの大きくなっている。このような木を2分探索木という。
- 2分探索木と新たな要素を与えられて、この要素を追加した木を返す関数  $\text{add\_bst} : \alpha \rightarrow \alpha \text{ tree} \rightarrow \alpha \text{ tree}$  を定義せよ。

# 課題5 (例1)

```
# add_bst 8 (Node(7,  
  Node(5, Node(4, Leaf, Leaf), Leaf),  
  Node(9, Leaf, Node(15, Leaf, Leaf))));;  
- : int tree =  
Node  
  (7, Node (5, Node (4, Leaf, Leaf),  
            Leaf),  
  Node (9, Node (8, Leaf, Leaf),  
        Node (15, Leaf, Leaf)))
```



# 課題5 (例2)



# 課題の提出方法

- To: ml-report@yl.is.s.u-tokyo.ac.jp
- Subject: Report 2 xxxxxx (学生証番号)
  - できるだけプログラムは  
本文に直接書いてください
- 〆切: 2002年4月30日(火) 13:00